

---

Subject: Replicating table 8.4: ETHIOPIA DHS 2000  
Posted by [Melese](#) on Sun, 18 Oct 2020 13:28:27 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Dear Expert, greetings. I tried to replicate table 8.4 "Perinatal mortality rate" using the 2000 Ethiopia DHS. "vcal\_1" doesn't exist the ETHIOPIA 2000 DHS data. Can anyone help with the STATA SYNTAX to calculate perinatal mortality rate using the ETHIOPIA 2000 data in the same way as mentioned in the GitHub: "https://github.com/DHSProgram/DHS-Indicators-Stata" "[https://github.com/DHSProgram/DHS-Indicators-Stata/tree/master/Chap08\\_CM](https://github.com/DHSProgram/DHS-Indicators-Stata/tree/master/Chap08_CM)"?

Regards  
Melese

---

---

Subject: Re: Replicating table 8.4: ETHIOPIA DHS 2000  
Posted by [Bridgette-DHS](#) on Tue, 20 Oct 2020 19:53:11 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Please see the following guidance from DHS Senior Technical Director: Trevor Croft:

See [https://userforum.dhsprogram.com/index.php?t=msg&th=5153&goto=9859&#msg\\_9859](https://userforum.dhsprogram.com/index.php?t=msg&th=5153&goto=9859&#msg_9859) for an existing discussion.

This also refers to [https://userforum.dhsprogram.com/index.php?t=msg&th=4692&goto=8657&#msg\\_8657](https://userforum.dhsprogram.com/index.php?t=msg&th=4692&goto=8657&#msg_8657) which provides code for Nepal, but the principle is the same for Ethiopia 2000.

See also the attached snippet of code.

Note that we don't have vcal\_1, but we have a pregnancy history, so can get the same information out of the pregnancy history.

### File Attachments

1) [Stillbirths Ethiopia 2000.do](#), downloaded 377 times

---

---

Subject: Re: Replicating table 8.4: ETHIOPIA DHS 2000  
Posted by [Melese](#) on Sun, 25 Oct 2020 13:10:18 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Hi Bridgette, greetings again and thank you for the reply. Unfortunately, I couldn't still generate the table 8.4: for EDHS 2000 which I easily did using EDHS 2005 and above data. I followed the discussions and failed to have the appropriate syntax. I used the following syntax with EDHS 2000 IR data file.

```
"keep caseid v001 v002 v003 v005 v008 v011 v013 v017 v018 v019 v021 v022 v023 v* s231ac_* s231c_*
```

```
rename *_0* *_*
reshape long s231ac_ s231c_, i(caseid) j(i)
drop if s231ac_==.
keep if s231ac_ > v008-60 & s231c>=7
rename s231ac_ b3"
```

merging with BR file using v001 v002 v003 creates error.

Can you help with fixing the merging and syntax for generating the variables "stillbirths", "early neonatal mortality", and "perinatal mortality", please?

Kindest regards  
Melese

---

Subject: Re: Replicating table 8.4: ETHIOPIA DHS 2000

Posted by [Melese](#) on Fri, 01 Jan 2021 04:42:01 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hi Bridgette, greetings again. I have been waiting for your responses for this long time. I have strictly followed the discussions so far but none of them explicitly mention some syntax to replicate table 8.4 from EDHS 2000. Example:- which variable should I use to replace the vcal\_1 (which is important to calculate the perinatal mortality rate) in the IR data file, s231ac\_1 or s231c\_1?? I want to calculate stillbirths, early neonatal mortality and perinatal mortality rates at the same time using the 2000 EDHS data. Any help, please?

Regards.

---

Subject: Re: Replicating table 8.4: ETHIOPIA DHS 2000

Posted by [Trevor-DHS](#) on Mon, 04 Jan 2021 17:57:42 GMT

[View Forum Message](#) <> [Reply to Message](#)

You nearly had it with your code! A couple of fixes for you:

- 1) You don't need the rename following the keep statement as there are less than 10 entries in the non-live pregnancy history, and thus no cases with a leading 0.
- 2) You had successfully isolated the stillbirths, but you also needed to drop the cases that were missing on s231c\_

I added in code to extract just the births in the last 5 years from the birth history (at the beginning), and then added code to combine the files (at the end) using append (not merge as you are appending births and stillbirths, not merging them together).

```
* select cases and variables of interest from the birth history
use "ETBR41FL.DTA", clear
```

```
keep caseid v001 v002 v003 v005 v008 v011 v013 v017 v018 v019 v021 v022 v023 b3 b6
```

```
* keep births in the last 5 years
```

```
keep if b3 > v008-60
```

save births, replace

\* select cases and variables of interest from the non-live pregnancy history  
use "ETIR41FL.DTA", clear

```
keep caseid v001 v002 v003 v005 v008 v011 v013 v017 v018 v019 v021 v022 v023 s231ac_*  
s231c_*
```

\* convert to long format to match birth history  
reshape long s231ac\_ s231c\_, i(caseid) j(i)

\* keep only pregnancies in the last 5 years of 7 month duration or longer  
drop if s231ac\_==.  
keep if s231ac\_ > v008-60 & s231c\_ >=7 & s231c\_ != .

\* rename cmc date of pregnancy to match birth history  
rename s231ac\_ b3

\* combine births and non-live pregnancies of 7+ months  
append using births

\* set weight variable  
gen wt = v005/1000000

\* capture stillbirths - those with a duration of pregnancy in s231c\_  
gen stillbirth = (s231c\_ != .)  
tab stillbirth [iw=wt]

\* capture early neonatal deaths (deaths in the first 7 days [0-6] = b6 < 107)  
gen earlyneonatal = (b6 < 107)  
tab earlyneonatal [iw=wt]

\* perinatal mortality is stillbirths or early neonatal deaths  
gen perinatal = (stillbirth | earlyneonatal)  
tab perinatal [iw=wt]

This shows 234 stillbirths, 421 early neonatal deaths, and a perinatal mortality rate of 5.25 percent, equivalent to 52.5 per thousand (the last digit is slightly different and I don't have an explanation why).

---

Subject: Re: Replicating table 8.4: ETHIOPIA DHS 2000  
Posted by [Melese](#) on Wed, 06 Jan 2021 11:53:01 GMT  
[View Forum Message](#) <> [Reply to Message](#)

Trevor, thank you very much. Thank you for the help to fix my questions. Thank you very much.

---